

# Introdução à Ciência da Computação

## Estruturas de Controle – Parte I

Prof. Ricardo J. G. B. Campello

## Créditos

Parte dos slides a seguir foram adaptados dos originais de A. L. V. Forbellone e H. F. Eberspächer

## Aula de Hoje

- Estrutura Sequencial
- Estruturas de Seleção
  - Simples
  - Composta
  - Encadeada
  - Múltipla Escolha

## Estrutura Sequencial

- O **fluxo de execução** (ou fluxo de controle) segue a mesma sequência linear da nossa escrita, ou seja:
  - De cima para baixo;
  - Da esquerda para direita
- Cada ação é seguida de um “;”
  - Objetiva separar uma ação da outra
  - Indica que a próxima ação da sequência deve ser executada

# Estrutura Sequencial

## Algoritmo 3.2 - Média Aritmética

```
início  
  // declaração de variáveis  
  real: N1, N2, N3, N4, // notas bimestrais  
    MA; // média anual  
  // entrada de dados  
  leia (N1, N2, N3, N4);  
  // processamento  
  MA ← (N1 + N2 + N3 + N4) / 4;  
  // saída de dados  
  escreva (MA);  
fim.
```

# Estrutura Sequencial

## Cálculo das Raízes Reais de $A \cdot x^2 + B \cdot x + C = 0$

```
início  
  // declaração de variáveis  
  real: A, B, C, // coeficientes da equação  
    D, // delta  
    X1, X2; // raízes  
  // entrada de dados  
  leia (A,B,C);  
  // processamento  
  D ← pot(B,2) - 4*A*C;  
  X1 ← (- B + rad(D)) / (2*A); // assumida real  
  X2 ← (- B - rad(D)) / (2*A); // assumida real  
  // saída de dados  
  escreva ("Primeira raiz = ", X1);  
  escreva ("Segunda raiz = ", X2);  
fim.
```

# Estrutura Sequencial

Cálculo do Volume de uma Esfera de Raio R

```
início
// declaração de variáveis
real: R, V; // raio e volume
// entrada de dados
leia (R);
// processamento
 $V \leftarrow (4/3) * 3.141592 * \text{pot}(R,3);$ 
// saída de dados
escreva ("Volume = ", V);
fim.
```

## Estruturas de Seleção

- Permitem alterar o fluxo de execução, de forma a selecionar qual parte deve ser executada
- Essa “decisão” de execução é tomada a partir de uma **condição**, que pode resultar apenas em **Verdade** (também V ou 1) ou **Falsidade** (F ou 0)
- Por esta razão são também denominadas de **estruturas condicionais**
- Uma condição é representada por expressões relacionais ou lógicas
- As estruturas de seleção podem ser classificadas em **simples**, **compostas** ou **encadeadas**

## Seleção Simples

```
se (condição) então  
  início // início do bloco verdade  
    comando 1;  
    comando 2;  
    ...  
    comando n;  
  fim; // fim do bloco verdade  
fimse;
```

- Quando a **condição** for verdadeira o “bloco verdade” é executado
- Quando a **condição** for falsa o “bloco verdade” **não** é executado
- Quando existe apenas 1 comando a ser executado, os delimitadores de bloco podem ser omitidos

## Seleção Simples

Algoritmo 3.4 - Média Aritmética com Aprovação

```
início  
  // declaração de variáveis  
  real: N1, N2, N3, N4, // notas bimestrais  
    MA; // média anual  
  // entrada de dados  
  leia (N1, N2, N3, N4);  
  // processamento  
  MA ← (N1 + N2 + N3 + N4) / 4;  
  // saída de dados  
  escreva (MA);  
  se (MA >= 7) então  
    escreva (“Aluno Aprovado !”); // 1 instrução: bloco é desnecessário  
  fimse;  
fim.
```

## Seleção Simples

### Exercício:

- Faça um algoritmo que tome as 2 notas de prova bimestrais (N1 e N2) de um aluno, mais as suas 2 notas de trabalhos (T1 e T2) e seu percentual de presença (P) e calcule a média final segundo o critério:
  - A nota final de provas MP é dada por uma **média ponderada**, sendo a primeira prova com peso 4 e a segunda com peso 6
  - A nota final de trabalhos (MT) é dada por uma **média aritmética**
  - A nota final do aluno é calculada por uma **média aritmética** das notas de prova (MP) e trabalho (MT)
  - Caso o aluno tenha 100% de presença, meio ponto de bonificação é adicionado à sua nota final, desde que não ultrapasse o teto de 10
- Escreva que o aluno está aprovado caso sua nota final seja igual ou superior a 5 e ele tenha ao menos 70% de presença
- Escreva que o aluno está reprovado caso sua nota final seja inferior a 5 ou ele tenha menos de 70% de presença

## Seleção Composta

**se** (condição) **então**

**início** // *início do bloco verdade*

comando 1;

comando n;

**fim**; // *fim do bloco verdade*

**senão**

**início** // *início do bloco falsidade*

comando 1;

comando n;

**fim**; // *fim do bloco falsidade*

**fimse**;

- Quando a **condição** for verdadeira o “bloco verdade” é executado
- Quando a **condição** for falsa o “bloco falsidade” é executado

## Seleção Composta

Algoritmo 3.5 - Média Aritmética com aprovação e reprovação

```
início
// declaração de variáveis
real: N1, N2, N3, N4, // notas bimestrais
      MA; // média anual
leia (N1, N2, N3, N4);
MA ← (N1 + N2 + N3 + N4) / 4;
escreva (MA);
se (MA >= 7) então
  início
    escreva ("Aluno Aprovado !");
    escreva ("Parabéns !");
  fim;
senão
  início
    escreva ("Aluno Reprovado !");
    escreva ("Estude mais !");
  fim;
fimse;
fim.
```

## Seleção Composta

### Exercício:

- Faça um algoritmo que tome as 2 notas de prova bimestrais (N1 e N2) de um aluno mais as suas 2 notas de trabalhos (T1 e T2) e calcule a média final segundo o critério:
  - A nota final de provas MP é dada por uma **média ponderada**, sendo a primeira prova com peso 4 e a segunda com peso 6
  - A nota final de trabalhos (MT) é dada por uma **média aritmética**
  - A nota final do aluno é calculada por uma **média aritmética** das notas de prova (MP) e trabalho (MT) caso ambas sejam iguais ou maiores do que 5 **ou** ambas sejam inferiores a 5. Caso contrário, a nota final será a menor dentre as notas finais de prova e trabalho

## Seleção Simples x Composta

### Exercício:

- Sejam S e H respectivamente o sexo e a altura de uma pessoa e assumamos que o peso médio ideal de um indivíduo seja dado pelas seguintes fórmulas:
  - Homens:  $(72,7 * H) - 58$
  - Mulheres:  $(62,1 * H) - 44,7$
- 1. Utilizando **seleções simples**, faça um algoritmo que tome S e H como entradas e então calcule e escreva o peso médio ideal do indivíduo
- 2. Refaça o algoritmo acima utilizando uma **seleção composta**
- 3. Aponte uma ou mais vantagens na utilização da seleção composta

## Seleção Encadeada

- Ocorre quando uma seleção tem como ação uma outra seleção
- Uma seleção encadeada pode ser:
  - **Heterogênea**: Quando não é possível identificar padrão de comportamento
  - **Homogênea**: Quando é possível identificar padrão de comportamento
    - se – então – se: quando depois de cada **então** ocorre outro **se**
    - se – senão – se: quando depois de cada **senão** ocorre outro **se**

# Seleção Encadeada Heterogênea

## Algoritmo 3.6 – Tipos de Triângulo

```
início
inteiro: A, B, C; // tamanho dos lados
leia (A, B, C);
se (A<B+C) e (B<A+C) e (C<A+B) então
  se (A=B) e (B=C) então
    escreva ("Triângulo Equilátero");
  senão
    se (A=B) ou (B=C) ou (A=C) então
      escreva ("Triângulo Isósceles");
    senão
      escreva ("Triângulo Escaleno");
  fimse;
fimse;
senão
  escreva ("Estes valores não formam um triângulo");
fimse;
fim.
```

Note que a condicional se-senão é vista como um único comando, não como um bloco. Por isso os delimitadores início e fim do bloco foram omitidos

# Seleção Encadeada Heterogênea

## Exercício:

- Dado o algoritmo a seguir, onde  $C_1, C_2, \dots, C_6$  são comandos (ações) quaisquer, indique quais deles serão executados se:
  - $A = V, B = V, C = F$
  - $A = F, B = V, C = F$
  - $A = F, B = V, C = V$
- Quais os valores de A, B e C para que somente os comandos  $C_5$  e  $C_6$  sejam executados?
- Existem valores para que somente  $C_6$  seja executado? Justifique.

```
início
// declaração de variáveis
lógico: A, B, C;
// processamento de dados
se A então C1;
senão
  início
    se B então
      se C então C2;
    senão
      início
        C3;
        C4;
      fim;
    fimse;
  fimse;
  C5;
fimse;
C6;
fim.
```

PS. Note a forma das condições para valores lógicos!

## Seleção Encadeada Heterogênea

### Exercício:

- Dado o algoritmo a seguir, onde  $C_1, C_2, \dots, C_8$  são comandos (ações) quaisquer, monte uma tabela com todas as possíveis combinações de valores lógicos de A, B e C e os comandos que serão executados para cada uma dessas combinações.

```

início
lógico: A, B, C;
se A então
  início
  C1; C2;
  fim;
senão
  início
  se B então
    início
    C4;
    se C então C5;
    senão C6;
    fimse;
  fim;
  fimse;
  C7;
  fim;
fimse;
C8;
fim.
    
```

PS. Note a forma das **condições** para valores lógicos!

## Seleção Encadeada Homogênea

```

se – então – se
se (Cond1) então
se (Cond2) então
se (Cond3) então
se (Cond4) então W;
fimse;
fimse;
fimse;
fimse;
    
```

Cond1	Cond2	Cond3	Cond4	Ação
V	V	V	V	W

É equivalente a:

```

se (Cond1) e (Cond2) e (Cond3) e (Cond4) então W;
fimse;
    
```

## Seleção Encadeada Homogênea

### Exemplo:

- Construa um algoritmo que receba como entradas o sexo, a idade e tempo de contribuição ao INSS, os últimos dois em anos, e escreva se essa pessoa tem direito a se aposentar, supondo que as regras de aposentadoria são:
  - 25 anos de contribuição e mínimo de 60 anos para mulheres
  - 30 anos de contribuição e mínimo de 65 anos para homens
- Escreva o algoritmo de duas formas diferentes, utilizando a estrutura encadeada homogênea do tipo se – então – se apresentada anteriormente e sua forma equivalente

## Seleção Encadeada Homogênea

### Solução 1:

```
início
  // declaração de variáveis
  caractere: sexo;
  inteiro: idade, tempo;
  // entrada de dados
  leia (sexo, idade, tempo);
  // processamento e saída de dados
  se (sexo = "M") então
    se (idade >= 65) então
      se (tempo >=30) então
        escreva ("Tem direito a se aposentar!");
      fimse;
    fimse;
  fimse;
  se (sexo = "F") então
    se (idade >= 60) então
      se (tempo >=25) então
        escreva ("Tem direito a se aposentar!");
      fimse;
    fimse;
  fimse;
fim.
```

## Seleção Encadeada Homogênea

Solução 2:

```
início
// declaração de variáveis
caractere: sexo;
inteiro: idade, tempo;
// entrada de dados
leia (sexo, idade, tempo);
// processamento e saída de dados
se (sexo = "M") e (idade >= 65) e (tempo >=30) então
    escreva ("Tem direito a se aposentar!");
fimse;
se (sexo = "F") e (idade >= 60) e (tempo >=25) então
    escreva ("Tem direito a se aposentar!");
fimse;
fim.
```

## Seleção Encadeada Homogênea

Solução Adicional:

```
início
// declaração de variáveis
caractere: sexo;
inteiro: idade, tempo;
// entrada de dados
leia (sexo, idade, tempo);
// processamento e saída de dados
se (sexo = "M") e (idade >= 65) e (tempo >=30) ou
    (sexo = "F") e (idade >= 60) e (tempo >=25) então
    escreva ("Tem direito a se aposentar!");
fimse;
fim.
```

PS. Seria preciso parênteses para indicar a ordem de precedência correta dos operadores lógicos de conjunção **e** e disjunção **ou** ?

## Seleção Encadeada Homogênea

```

se X=V1 então
  C1;
fimse;
se X=V2 então
  C2;
fimse;
se X=V3 então
  C3;
fimse;
se X=V4 então
  C4;
fimse;

```

```

se X=V1
  então C1;
  senão se X=V2
    então C2;
    senão se X=V3
      então C3;
      senão se X=V4
        então C4;
        fimse;
      fimse;
    fimse;
  fimse;

```

X=V1	X=V2	X=V3	X=V4	Ação
V	F	F	F	C1
F	V	F	F	C2
F	F	V	F	C3
F	F	F	V	C4

X=V1	X=V2	X=V3	X=V4	Ação
V	-	-	-	C1
F	V	-	-	C2
F	F	V	-	C3
F	F	F	V	C4

Alguns  
Vantagem  
?



## Seleção Encadeada Homogênea

### Exemplo:

- Construa um algoritmo que receba a idade de um nadador e classifique-o em uma das seguintes categorias:
  - 5 até 8 anos: Infantil
  - 9 até 13 anos: Juvenil A
  - 14 até 17 anos: Juvenil B
  - Maiores de 18: Adulto
- Escreva o algoritmo de duas formas diferentes, utilizando a estrutura encadeada homogênea se - senão - se apresentada anteriormente e sua forma equivalente com múltiplas seleções simples se - então.

# Seleção Encadeada Homogênea

Solução  
1:

```
início
// declaração de variáveis
inteiro: idade;
// entrada de dados
leia (idade);
// processamento e saída de dados
se (idade >=5) e (idade <= 8) então
    escreva ("Infantil");
fimse;
se (idade >=9) e (idade <=13) então
    escreva ("Juvenil A");
fimse;
se (idade >=14) e (idade <=17) então
    escreva ("Juvenil B");
fimse;
se (idade >=18) então
    escreva ("Adulto");
fimse;
fim.
```

# Seleção Encadeada Homogênea

Solução  
2:

```
início
// declaração de variáveis
inteiro: idade;
// entrada de dados
leia (idade);
// processamento e saída de dados
se (idade >=5) e (idade <= 8) então
    escreva ("Infantil");
senão
    se (idade >=9) e (idade <=13) então
        escreva ("Juvenil A");
    senão
        se (idade >=14) e (idade <=17) então
            escreva ("Juvenil B");
        senão
            se (idade >=18) então
                escreva ("Adulto");
            fimse;
        fimse;
    fimse;
fimse;
fim.
```

## Seleção Encadeada Homogênea

### Exercício:

- A partir da idade informada de uma pessoa, elabore um algoritmo que informe a sua classe eleitoral, sabendo que menores de 16 não votam (não votantes), que o voto é obrigatório para adultos entre 18 e 65 anos (eleitor obrigatório) e que o voto é opcional para eleitores com 16, 17 ou mais que 65 anos (eleitor facultativo).
- Escreva o algoritmo de duas formas diferentes:
  - Utilizando múltiplas seleções simples se - então
  - Utilizando uma estrutura encadeada homogênea se - senão - se

## Seleção de Múltipla Escolha

- Seleções encadeadas homogêneas se-senão-se são bastante frequentes para o tratamento de listas de valor
- Para simplificar a escrita, pode-se utilizar uma seleção de múltipla escolha:

**escolha X**

**caso V1:** C1;

**caso V2:** C2;

**caso V3:** C3;

**caso V4:** C4;

**caso contrário:** C5;

**fimescolha;**

## Seleção de Múltipla Escolha

### Algoritmo 3.7 – Múltipla Escolha

**início**

**real:** Preço;

**inteiro:** Origem;

**leia** (Preço, Origem);

**escolha** Origem

**caso 1: escreva** (“Valor =”, Preço, “ – Produto do Sul”);

**caso 2: escreva** (“Valor =”, Preço, “ – Produto do Norte”);

**caso 3: escreva** (“Valor =”, Preço, “ – Produto do Leste”);

**caso 4: escreva** (“Valor =”, Preço, “ – Produto do Oeste”);

**caso contrário: escreva** (“Valor =”, Preço, “ – Produto importado”);

**fimescolha;**

**fim.**

## Seleção de Múltipla Escolha

### Exercício:

- Utilizando seleção de múltipla escolha, escreva um algoritmo que, a partir de um mês fornecido (número inteiro de 1 a 12), apresente o nome dele por extenso ou uma mensagem de mês inválido

## Seleção Encadeada x Múltipla Escolha

### Exercício:

- Dados os seguintes critérios de classificação de candidatos em uma prova:
  - A: 90 a 100 pontos
  - B: 70 a 89 pontos
  - C: 50 a 69 pontos
  - D: 30 a 49 pontos
  - E: 0 a 29 pontos
- Faça um algoritmo que receba o número de pontos de um candidato e imprima a sua classificação utilizando:
  - Múltiplas seleções simples
  - Seleção encadeada homogênea se – senão – se

## Exercícios Adicionais

- ◆ Capítulo III de (Forbellone & Eberspächer, 2005):
  - Estudar exercícios de fixação
  - Resolver exercícios propostos